
_INFILE_

# in a data step with an INPUT statement, SAS creates

an input buffer where it holds your data prior to

moving the values of variables into the program data

vector  

# you can access the contents of that buffer using the

variable name _INFILE_

# _INFILE_ is an automatic variable whose value is

accessible within a data step but is not output to any

data set being created in the data step

_INFILE_

data names;
input @; â
_infile_ = upcase(_infile_); ã
input name :$10. age city :$10. state :$2.; ä
datalines;
mike 25 albany ny
Sara 15 Washington DC
;

an INPUT statement with no variable names places a record from

the DATALINES file into the input buffer â ... the contents of the

buffer is named _INFILE_ and all text in the buffer is converted to

uppercase ã ... variables are read from the contents of the input

buffer ä that was "held" by the @ â in the first INPUT statement 

_INFILE_

original data ...   data set ...
datalines;
mike 25 albany ny
Sara 15 Washington DC
;

# in addition to SAS documentation, there are several
papers describing uses of _INFILE_

# output buffer and _FILE_ discussed in SAS
documentation, but cannot find any papers that
describe possible uses of _FILE_ ... examples are used
to show some "possibilities"

_FILE_

EXAMPLE #1 ... add a variable to a data set showing

which values of numeric variables are below or at/above

the median

data set SASHELP.CLASS has three numeric variables ...

age (years), height (inches), weight (pounds) ... find

median values for MALES

proc means data=sashelp.class

maxdec=2 median;

where sex eq 'M';

run;

_FILE_

proc format; â

value ag low-<13.5   = '0'  other = '1';

value ht low-<64.15  = '0'  other = '1';

value wt low-<107.25 = '0'  other = '1';

run;

filename nosee dummy; ã

three FORMATS are created that will divide values into those

below and those at/above the median â ... a FILENAME

statement assigns the FILEREF NOSEE to the device type DUMMY

ã (use of DUMMY specifies that any output written to the

FILEREF NOSEE is discarded) 

_FILE_

data males (drop=sex);

file nosee; â

set sashelp.class (where=(sex eq 'M')); ã

put age ag. height ht. weight wt. @; ä

aghtwt = _file_; å

put; æ

run;

a FILE statement directs the output from PUT statements to

FILEREF NOSEE â ... observations for MALES are read ã ... a PUT

statement with a trailing @ writes the formatted values of

variables to the output buffer ä (the @ holds the values in  the

buffer) ... the contents of the buffer has the variable name _FILE_

and it is assigned to the variable AGHTWT å ... the buffer is

cleared with PUT æ

_FILE_ 

data set MALES

Thomas is below the median for all

three variables, Robert is at/above

the median for both HEIGHT and

WEIGHT,  Henry is at/above the

median for only AGE, William is

above the median for all three

variables, etc.

# "aha" moment ... PUT with @ and _FILE_ allow you to

easily CONCATENATE the FORMATTED VALUES of

variables

8



_FILE_

what is the LENGTH of the new

variable AGHTWT  ...

aghtwt = _file_;

the new variable is the 

same length as that of the output buffer

when creating a new variable using _FILE_ you should

add a LENGTH statement to the data step ...

length aghtwt $3;

_FILE_

without the second PUT (the one

without the @), the output buffer

is never cleared and the PUT

statement with an @ keeps

adding values to the output buffer

(the variable _FILE_) 

data males (drop=sex);

file nosee; 

set sashelp.class (where=(sex eq 'M')); 

put age ag. height ht. weight wt. @;

aghtwt = _file_; 

run;

_FILE_

same result, less SAS code ...

data males (drop=sex);

set sashelp.class (where=(sex eq 'M'));

put @1 age ag. height ht. weight wt. @; â

aghtwt = _file_;

run;

the data step has NO FILE statement, thus all PUT statements

write to the LOG ... however, a PUT statement with an @ holds

the PUT statement results in the output buffer and does not write

to the LOG â ... there is NO second PUT statement without the @

to clear the buffer since the PUT @1 always writes values to

columns 1 through 3 â 

_FILE_

look at the log ...
344  data males (drop=sex);

345  set sashelp.class (where=(sex eq 'M'));

346  put @1 age ag. height ht. weight wt. @;

347  aghtwt = _file_;

348  run;

111     one line written to the LOG (from the last
observation ... clears the output buffer)

NOTE: There were 10 observations read from the

data set SASHELP.CLASS. WHERE sex='M';

NOTE: The data set WORK.MALES has 10 observations

and 5 variables.

data set ANSWERS

FIND, WHICHC, IN OPERATOR

EXAMPLE #2 ... searching for variable values

simple task ... given data

set ANSWERS ... find

observations with at least

one answer that is "Y"

FIND, WHICHC, IN OPERATOR

data atleast1y;

set answers;

if find(catt(of q:),'Y');

run;

data atleast1y;

set answers;

if whichc('Y', of q:);

run;

data atleast1y;

set answers;

array q(10);

if 'Y' in q;

run;

concatenate values of

variables q1-q10 and use

FIND to search for "Y"

use WHICHC function see

if any value of q1-q10 is

equal to "Y"

use an IN operator to

search an array (values

q1-q10) for "Y" 

data set DIAGNOSES

IN OPERATOR+COLON MODIFIER

more complex task ... given data

set DIAGNOSES ... find

observations with at least one

diagnosis that STARTS with the

string "250" (diabetes) ... complex 

with CAT and FIND functions,

cannot use WHICHC, easy with IN

operator

data diabetes;

set diagnoses;

array dx(5);

if '250' in : dx;

run;

the colon modifier after the

IN operator limits the search

to the first three characters

of the various diagnoses

_FILE_

task common to both simple search (look for variables

with a value of "Y") and more complex search (look for

variables with a value that starts with "250") ... SEARCH

MANY VARIABLES for a SINGLE VALUE

what about SEARCHING MANY VARIABLES for MANY

VALUES, the equivalent of ...

if <many values> in <many variables>;

data set DIAGNOSES ... find TBI

_FILE_

search for diabetes was a search

for one value ... "250"

search for traumatic brain injury

(TBI) is a search for multiple

values ... "800"-"80199", 

"803"-"80499" , "850"-"85419",  "9501"-"95039",

"95901", "99555" ... for the first observation in data set

DIAGNOSES, that would look like ...

if <95901, 78039, 4280, 87342, 81612> in

<800-80199,  803-80499 , 850-85419,  9501-95039,

95901, 99555>;

_FILE_

solution with a FORMAT, _FILE_, and FIND

first, create a FORMAT with ranges and individual values

that indicate TBI ...

proc format;

value $tbi

'800'-'80199', '803'-'80499' , '850'-'85419', 

'9501'-'95039', '95901' , '99555' = '1'

other = '0' ;

run;

18



data set TBI

_FILE_

next, use the FORMAT in a data step ...

data tbi;

set diagnoses;

put @1 (dx1-dx5) ($tbi.) @; â

if find(_file_,'1'); ã

run;

a PUT statement writes a string of 1s and 0s to the

output buffer (formatted values of the diagnoses, 1

indicates TBI) â ... a FIND function looks for 1s in the

output buffer ã

_FILE_

EXAMPLE #3 ... search for variable values (TBI) and add a
variable (values 1, 0, X) that indicates if a diagnosis is
TBI, not TBI, or missing ... a combination of examples #1
and #2

proc format;

value $tbi

'800'-'80199', '803'-'80499' , '850'-'85419', 

'9501'-'95039', '95901' , '99555' = '1' â

other = '0' ã

' ' ='X'; ä

run;

FORMAT differentiates among TBI â, not TBI ã, and
missing ä

_FILE_

data tbi;

length tbi $5; â

set diagnoses;

put @1 (dx1-dx5) ($tbi.) @; ã

if find(_file_,'1'); ã

tbi = _file_; ä

run;

a LENGTH statement sets the length of the new variable TBI â ... a
PUT statement writes a string of 1s and 0s to the output buffer
and FIND locates TBI ã ... a new variable is created ä

locations of TBI diagnoses
indicated with 1s in variable TBI 

CONCLUSION

# VARIOUS USES OF THE CONTENTS OF THE INPUT
BUFFER CREATED WITH AN INPUT STATEMENT AND
ACCESSED  VIA THE VARIABLE _INFILE_ HAVE BEEN
SHOWN IN SEVERAL PAPERS

# THIS PRESENTATION (AND PAPER) DEMONSTRATE
THAT THE CONTENTS OF THE OUTPUT BUFFER
CREATED WITH A PUT STATEMENT CAN BE ACCESSED
VIA THE VARIABLE _FILE_

# SEVERAL EXAMPLES DEMONSTRATED HOW USEFUL
THE VARIABLE _FILE_ CAN BE (AND THERE ARE MORE
USES SHOWN IN THE PAPER)

